Appendix D
Computer Program for Designing Banked Curves for Supercritical Flow in Rectangular Channels

D-1. Introduction

a. General. The design of curves for supercritical flow may include several alternatives which produce curves that perform satisfactorily for the design flow and that are compatible with existing field conditions. The solution for any one alternative is time consuming, requiring trial-and-error computations. The alternative designs described in this appendix include basic limiting design criteria developed by the US Army Engineer District (USAED), Los Angeles. Combining the results of two or more of these alternatives should produce a satisfactory design for nearly any condition. A list of symbols used in the program (Plate D-1), a program listing (Plate D-2), a program flow chart (Plate D-3), subroutine flow charts (Plates D-4 and D-5), an example input sheet (Plate D-6), and an example output listing (Plate D-7) are included herein. The computer program is written in FORTRAN IV and has been tested on a GE-425 computer through a remote teletype terminal.

b. Hydraulic elements. The hydraulic elements are computed using an equation for open channel flow adapted from the Colebrook-White equation for pipe flow (HDC 224-1). The equivalent open channel flow equation in terms of Chezy C is

$$C = -32.6 \log_{10}\left(\frac{C}{5.2R_n} + \frac{k}{12.2R}\right) \qquad \text{(D-1)}$$

where

R_n = Reynolds number = 4RV/ν

R = hydraulic radius

V = velocity

ν = kinematic viscosity of water at given temperature

k = assigned equivalent roughness height

Equation D-1 is graphically presented in Plate 3, Appendix B. Its derivation is described in HDC 631 to 631-2. The equation has been used in the program subroutine because it is equally applicable to all flow zones and eliminates the need of advanced prediction of the channel flow type.

c. Spiral transition. The modified spiral (McCormick 1948) is used for the transition between the tangent and fully banked sections of the curved channel.[1] This type of curve permits location of the channel interior and exterior walls by means of a simple coordinate system based upon a series of circular arcs of uniform length compounded to approximate a conventional spiral. The initial arc has a large radius, and the radius of each succeeding arc is decreased in a prescribed manner until the desired channel curve radius is attained. The advantage of the modified spiral over a conventional spiral is realized during field layout of the short chord lengths required for the concrete wall forms.

d. Tables of spiral transition. Tables have been prepared for 22 different spirals (McCormick 1948) to facilitate design layout and field location. The curve numbers in the tables correspond to the number of seconds in the central angle of the first arc of the spiral. This designation is followed in the computer program. However, the curve and corresponding number computed by the program may not be listed in the modified spiral tables because the program selects the exact curve for the specified radius and spiral length.

D-2. Description of Problem

The basic criteria for the design of spiral-banked curves for rectangular channels are given by Equations 2-33, 2-34, and 2-36 of the main text. A review of these equations reveals that the designer has several alternatives at his disposal to satisfy the design criteria. For example, if the minimum radius of curvature is selected from Equation 2-34, i.e.,

$$r_{min} = \frac{4WV^2}{gy} = 4WF^2 \qquad \text{(2-34, D-2)}$$

then the maximum allowable amount of banking (difference between inside and outside invert elevations in the circular curve) is required. The amount of banking (e ,

[1] References cited in this appendix are included in Appendix A.

Plate D-8) is equal to twice the superelevation given by Equation 2-36. For C = 0.5 ,

$$e = 2\Delta y = (2)(0.5)\frac{V^2W}{gr} = \frac{V^2W}{gr} \qquad \text{(2-36, D-3)}$$

Where $r = r_{min}$, e is therefore a maximum, or

$$e_{max} = \frac{V^2W}{gr_{min}} = \frac{y}{4} \qquad \text{(D-4)}$$

Also, the minimum recommended spiral length for banked curves (Equation 2-33) is

$$L_s = 30\Delta y \qquad \text{(2-33, D-5)}$$

The choice of minimum radius of curvature in Equation 2-34 (D-2), maximum banking (Equation D-3), and the corresponding spiral lengths (Equation D-5) results in the shortest total curve length. If radii greater than minimum are selected, then according to Equation 2-36 (D-3), the amount of banking would be less than that expressed by Equation D-4. Moreover, both the radius of curvature and the spiral lengths may be arbitrarily selected to satisfy field conditions so long as they exceed the minimum criteria as expressed by Equations 2-33 (D-5) and 2-34 (D-2). Also, the entering and exit spiral lengths do not have to be equal as long as each exceeds the value determined by Equation 2-33 (D-5). It should be noted that with banked inverts, an upper limit on the radius of curvature exists at which the banking (2Δy) is less than 0.5ft. In this case banking and spiral transitions may not be necessary (paragraph 2-5b). Substituting this limiting (0.5 ft) value for e into Equation D-3 and solving for r , the limits for the radius of curvature where banking is required can be expressed as

$$4WF^2 \le r \le 2WyF^2 \qquad \text{(D-6)}$$

Lastly, the transverse slope 2Δy/W of the water surface should not exceed 0.18 which corresponds to a slope angle ϕ of 10 deg (Equation 2-36, D-3).

a. Free drainage. Another criterion that must be satisfied in some cases is that the channel be free draining. Banking is introduced by rotation of the bottom about the channel invert center line. Therefore, to provide free drainage along the inside wall, the product of the exit spiral length and centerline invert slope must be greater than the superelevation (Δy), i.e.

$$L_sS > \Delta y \qquad \text{(D-7)}$$

Generally, the curves designed for minimum radii (Equations D-2, D-4, and D-5) will not be free draining unless the channel center-line invert slope is extremely steep. There are several ways of accomplishing free drainage by varying independently or dependently the spiral length and channel invert slope. However, the most common method is illustrated in Plate D-8. In this plate the length of the exit spiral is increased to satisfy Equation D-7 while the channel invert slope is held constant. The unequal spiral lengths generated by increasing the exit spiral should perform satisfactorily, but if symmetry is desired, the entering spiral may be equally increased.

b. Alternatives. The following list of design alternatives is based on the previously discussed criteria.

(1) Minimum radius of curvature (Equation D-2), maximum banking (Equation D-4), and corresponding spiral length (Equation D-5). Shortest total length. Not free draining. Equal spiral lengths.

(2) Minimum radius of curvature (Equation D-2), maximum banking (Equation D-4), and arbitrary spiral length greater than value given by Equation D-5. Not free draining. Equal spiral lengths.

(3) Arbitrary radius of curvature greater than Equation D-2, banking in accordance with Equation D-3, and corresponding spiral length (Equation D-5). Not free draining. Equal spiral lengths.

(4) Arbitrary radius of curvature and spiral length both greater than value given by Equations D-2 and D-5, respectively. Banking per Equation D-3. Not free draining. Equal spiral lengths.

(5) Arbitrary radius of curvature greater than value given by Equation D-2. Arbitrary entering and exit spiral lengths (unequal) but both greater than value given by Equation D-5. Banking computed using Equation D-3. Not free draining. Unequal spiral lengths.

(6) Same as (1) above except free drainage provided by increasing exit spiral length. Entering spiral remains per Equation D-5. Unequal spiral lengths.

(7) Same as (1) above except free drainage provided by increasing length of both spirals. Equal spiral lengths.

(8) Same as (3) above except free drainage provided by increasing exit spiral length; entering spiral remains per Equation D-5. Unequal spiral lengths.

(9) Same as (3) above except free drainage provided by increasing length of both spirals. Equal spiral lengths.

(10) Same as (4) above except free drainage provided by increasing exit spiral length. Entering spiral length retains arbitrary assigned value. Unequal spiral lengths.

(11) Same as (4) above except free drainage provided by increasing lengths of both spirals. Equal spiral lengths.

(12) Same as (5) above except free drainage provided by increasing exit spiral length. Unequal spiral lengths.

The various characteristics of these alternatives are compared in Plate D-9.

D-3. Description of Program

The program herein described is comprehensive in that any of the above-listed alternatives can be solved. The program is written for remote terminal use because of the increasing use of remote terminals and the definite advantages gained through this mode of operation provided the volume of input-output data is moderate. The main advantage of the remote terminal is that the program can be written so that it is user oriented. The user is guided by typewritten messages throughout the program execution, and the program is controlled by the user's response to these typed questions. Communication between the user and the computer during program solutions results in advantages in problems having alternative solutions. Conversion to batch processing is relatively simple and only requires modification of the READ statements in the program. A complete description of each input variable is given prior to its respective READ statement in the program listing (Plate D-2).

D-4. Input Data

a. Hydraulic parameters. Plate D-6 shows sample input data format. The first line of input represents the given design data, which include the discharge (cfs), channel center-line invert slope (ft/ft), channel width (ft), equivalent roughness height (ft), water temperature (^oF), and the deflection angle (deg) between the curve tangents. Since the hydraulic elements are solved by trial and error using Equation D-1, the roughness parameter is the equivalent roughness height k. The curve design should be based on the maximum average channel velocity, for which the recommended minimum value of k for concrete-lined channels is 0.002 ft (paragraph 2-2c). The k value should always be the lowest value of the expected equivalent roughness height range if the minimum of that range is less than 0.002 ft. However, the wall heights in the curve, as in the case of the straight channel, should be designed for capacity based on $k = 0.007$ ft (paragraph 2-2c) or a higher value if anticipated.

b. Circular curve data. The second line of input is the design radius for the circular curve. The recommended minimum radius as calculated from the given flow conditions (Equation D-2) is stated in the typed request for this variable. If the minimum radius is desired, then *0.0* is assigned, otherwise, the desired value is typed in. The third and fourth lines of input are for the entering and exit spiral lengths, respectively. Similar to the request for the radius, the minimum spiral length based on Equation 2-33 (D-5) is stated. Either *0.0* or the desired value for each spiral length is assigned.

c. Radius of curvature. Occasionally, field conditions will limit the radius of curvature such that it must be less than the recommended minimum. The program can design a curve for values of radius and spiral length that are less than the recommended minimums, but the amount of banking will exceed the value given by Equation D-4. Furthermore, the cross-slope angle of the water surface will be greater than that which would occur with the recommended minimum radius. Should it exceed 10 deg, a message will be generated to advise the user that this criterion has been violated. Model testing of curves that violate any of these criteria should be considered.

d. Free drainage. The fifth line of input is for providing free drainage. The question is typed on the keyboard, and the user replies "yes" or "no." If yes, the program computes the length of spiral necessary to provide free drainage and compares it with the exit spiral length as per input line 4. If the value supplied by input line 4 is less than the length required for free drainage, the exit spiral length is increased accordingly. Input line 6 affords the user the option to make both spirals equal length for symmetry and appears only if the reply of "yes" is made to input line 5.

D-5. Program Output

The program output (Plate D-7) consists of the hydraulic and geometric design of the channel curve. The hydraulic elements include a listing of all the given design data and the pertinent computed hydraulic parameters. The channel curve elements are presented in two parts. The first part gives the information required to prepare contract drawings. The second part gives the detailed data for field layout of the channel center line. The field book format is set up under the assumption that the entering spiral is first surveyed from TS; then the transit is moved to the end of the curve (ST) and the exit spiral backed in; finally, the transit is moved to the downstream end of the entering spiral (SC) and the circular curve surveyed. This is the recommended procedure for field layout found in most route survey textbooks. Curve stations are established using 12.5-ft chords around the spiral and 100-ft chords around the circular curve rather than the actual curve lengths. Shorter chord lengths may be required at the beginning and end of the circular curve, but these can be easily computed during the actual field layout.

- **A:** DEFLECTION ANGLE FOR POINTS ALONG SPIRAL, DEG
- **ANG:** DEFLECTION ANGLE BETWEEN INITIAL AND FINAL TANGENTS; TOTAL CENTRAL ANGLE OF CIRCULAR CURVE AND SPIRALS, DEG
- **ARC(1):** LENGTH OF ENTERING SPIRAL, FT
- **ARC(2):** LENGTH OF EXIT SPIRAL, FT
- **ARCF:** DIGITS TO RIGHT OF PLUS SIGN IN STATION NUMBER CORRESPONDING TO LENGTH OF ENTERING SPIRAL, I.E. 00+00.00
- **ARC1F:** DIGITS TO RIGHT OF PLUS SIGN IN STATION NUMBER CORRESPONDING TO LENGTH OF EXIT SPIRAL, I.E. 00+00.00
- **ARCL:** LENGTH OF ENTERING SPIRAL, FT
- **ARCL1:** LENGTH OF EXIT SPIRAL, FT
- **ARCL2:** MINIMUM LENGTH OF EXIT SPIRAL TO PROVIDE FREE DRAINAGE, FT
- **AREA:** CROSS-SECTION AREA OF FLOW IN CHANNEL, FT^2
- **B:** CHANNEL WIDTH, FT
- **C:** CHORD LENGTH OF INDIVIDUAL ARCS ALONG THE SPIRAL, FT
- **CHORDC:** CHORD LENGTH USED IN STAKING OUT CIRCULAR CURVE; IN THIS CASE 100 FT
- **CHZC:** CHEZY C AS CALCULATED BY CHEZY'S EQUATION
- **CHZC1:** CHEZY C AS CALCULATED BY COLEBROOK-WHITE TRANSITIONAL ZONE EQUATION
- **CLGTHE:** DIGITS TO RIGHT OF PLUS SIGN IN STATION NUMBER CORRESPONDING TO LENGTH OF CIRCULAR CURVE, I.E. 00+00.00
- **CSS:** POINT OF CHANGE FROM CIRCULAR CURVE TO SPIRAL
- **CSSF:** DIGITS TO RIGHT OF PLUS SIGN IN STATION NUMBER CORRESPONDING TO THIS POINT (C.S.), I.E. 00+00.00
- **DEFC:** DEFLECTION ANGLES FOR POINTS ALONG CIRCULAR CURVE, MIN
- **DEGC:** DEGREE OF CURVATURE OF CIRCULAR CURVE, DEG
- **DELTA:** CENTRAL ANGLE OF INDIVIDUAL ARCS, SEC
- **DELTAN:** MEDIAN ANGLE OF INDIVIDUAL ARCS, SEC
- **DELTA1:** CENTRAL ANGLE OF WHOLE SPIRAL, DEG
- **DELTAC:** CENTRAL ANGLE OF CIRCULAR CURVE, DEG
- **DIFF:** PERCENT DIFFERENCE BETWEEN CHZC AND CHZC1
- **FRUDN:** FROUDE NUMBER
- **G:** ACCELERATION DUE TO GRAVITY, FT/SEC^2
- **IARC:** NUMBER OF HUNDRED DIGITS IN ENTERING SPIRAL LENGTH, I.E., THE LENGTH OF SPIRAL IS BROKEN DOWN INTO THE FORM 00+00.00. THE HUNDRED DIGITS ARE THOSE TO THE LEFT OF THE + SIGN, FT
- **IARC1:** SAME AS IARC EXCEPT THAT IT DESIGNATES EXIT SPIRAL LENGTH, FT
- **ICSS:** DIGITS TO LEFT OF PLUS SIGN IN STATION NUMBER OF C.S., I.E. 00+00.00
- **ISCC:** SAME AS ICSS EXCEPT THAT IT DESIGNATES THE STATION NUMBER OF THE S.C.
- **ICLGTH:** SAME AS IARC EXCEPT THAT IT DESIGNATES THE LENGTH OF THE CIRCULAR CURVE
- **ISTA:** DIGITS TO LEFT OF PLUS SIGN IN STATION NUMBERS OF CIRCULAR CURVE
- **ISTT:** DIGITS TO LEFT OF PLUS SIGN IN STATION NUMBER OF S.T., I.E. 00+00.00
- **ISTAS:** DIGITS TO LEFT OF PLUS SIGN FOR ANY STATION NUMBER ON EXIT OR ENTERING SPIRAL
- **KS:** EQUIVALENT ROUGHNESS HEIGHT OF CHANNEL BOUNDARY, FT
- **KV:** KINEMATIC VISCOSITY OF WATER, FT^2/SEC
- **LS:** DESIGN EXIT SPIRAL LENGTH; MUST BE GREATER THAN 30 TIMES THE SUPER-ELEVATION, FT
- **LS1:** DESIGN ENTERING SPIRAL LENGTH; MUST BE GREATER THAN 30 TIMES THE SUPER-ELEVATION, FT
- **MDG, MIN, MSEC:** NUMBER OF DEGREES, MINUTES, AND SECONDS, RESPECTIVELY, IN DEFLECTION ANGLE TO ANY POINT ON SPIRAL
- **MGDB, MNIC, MSECB:** NUMBER OF DEGREES, MINUTES, AND SECONDS, RESPECTIVELY, IN BACK DEFLECTION ANGLE TO T.S. WITH TRANSIT AT S.C.
- **MDGC, MINC, MSECC:** NUMBER OF DEGREES, MINUTES, AND SECONDS, RESPECTIVELY, IN DEFLECTION ANGLE TO ANY POINT ON CIRCULAR CURVE WITH TRANSIT AT S.C.
- **Q:** DESIGN CHANNEL DISCHARGE, CFS
- **R:** RADIUS OF CURVATURE OF INDIVIDUAL ARCS ALONG SPIRAL, FT
- **RC:** DESIGN RADIUS OF CURVATURE, FT
- **RELRO:** RATIO OF HYDRAULIC RADIUS TO EQUIVALENT ROUGHNESS HEIGHT (RELATIVE ROUGHNESS)
- **REYN:** REYNOLDS NUMBER
- **RMIN:** MINIMUM RADIUS OF CURVATURE OF CIRCULAR CURVE, FT
- **S:** CHANNEL CENTER-LINE INVERT SLOPE, FT/FT
- **SCC:** POINT OF CHANGE FROM SPIRAL TO CIRCULAR CURVE MEASURED FROM BEGINNING OF CURVE (S.S.), FT
- **SCCF1:** DIGITS TO RIGHT OF PLUS SIGN IN STATION NUMBER OF THE S.C., I.E. 00+00.00
- **STA:** STATION NUMBERS OF POINTS ALONG CIRCULAR CURVE
- **STAF:** DIGITS TO RIGHT OF PLUS SIGN IN CIRCULAR CURVE STATION NUMBERS, I.E. 00+00.00
- **STAS:** STATION NUMBERS OF 12.5-FT-CHORD SPIRAL POINTS
- **STASF:** DIGITS TO RIGHT OF PLUS SIGN IN STATION NUMBERS OF SPIRAL POINTS, I.E. 00+00.00
- **STT:** POINT OF CHANGE FROM SPIRAL TO TANGENT OR S.T., FT
- **STTF:** DIGITS TO RIGHT OF PLUS SIGN IN STATION NUMBER OF S.T.
- **TEMP:** WATER TEMPERATURE, °F
- **TL:** TANGENT DISTANCE OF SPIRALED CURVE: DISTANCE FROM T.S. OR S.T. TO POINT OF INTERSECTION OF TANGENTS (PI), FT
- **VEL:** MEAN CHANNEL VELOCITY, FPS
- **VELH:** VELOCITY HEAD OF FLOW IN CHANNEL, FT
- **X:** CENTER-LINE COORDINATE OF POINTS ALONG SPIRAL ALONG PRIMARY TANGENT, FT
- **YC:** DEPTH OF FLOW IN CHANNEL, FT
- **Y:** ORDINATE OR TANGENT OFFSET OF POINTS ALONG SPIRAL, FT
- **Z:** AMOUNT OF BANKING, OR DIFFERENCE IN ELEVATION BETWEEN OUTSIDE AND INSIDE INVERT OF CHANNEL, FT

COMPUTER PROGRAM SYMBOLS

PLATE D-1

```
100C PROGRAM TO DESIGN A SPIRAL-BANKED CURVE FOR SUPERCRITICAL FLOW IN
110C A RECTANGULAR CHANNEL.  INPUT INCLUDES DESIGN DISCHARGE(Q) CFS,
120C CHANNEL CENTERLINE INVERT SLOPE(S) FT/FT, CHANNEL WIDTH(B) FT,
130C EQUIVALENT ROUGHNESS HEIGTH(KS) FT, WATER TEMPERATURE(TEMP) DEG-
140C F, DEFLECTION ANGLE BETWEEN TANGENTS(ANG) DEGREES, RADIUS OF
150C CURVATURE OF CIRCULAR CURVE(RC) FT, ENTERING SPIRAL LENGTH(LS1) FT,
160C EXIT SPIRAL LENGTH(LS) FT, AND THE INPUT OF YES OR NO TO THE
170C QUESTIONS OF FREE DRAINAGE AND EQUAL SPIRAL LENGTHS DESIRED?
180C     OUTPUT INCLUDES GIVEN DESIGN DATA,IMPORTANT HYDRAULIC PARAMETERS
190C AND A COMPLETE CENTERLINE LAYOUT INCLUDING DEFLECTION ANGLES FOR
200C FIELD LAYOUT.
210 REAL KS,KV,LS,LS1
220 G=32.2
230 PRINT 1
240 1 FORMAT(/,"READ DESIGN DATA - Q,S,B,KS,TEMP,ANG")
250 READ,Q,S,B,KS,TEMP,ANG
255 KV=(-.00000207*TEMP**3+.0006*TEMP**2-.0674*TEMP+3.535)/10**5
260 CALL HYD(Q,S,B,KS,KV,ANG,YC,HYRAD,CHZC1,VEL,FRUDN,REYN,RELRO,Z,
270&RMIN,ARCL)
280 PRINT 2,RMIN
290 2 FORMAT(/,"READ THE DESIGN RADIUS OF CURVATURE. THE MINIMUM REC
300&OMMENDED",/,"RADIUS =",F10.2,"FT. IF MINIMUM RADIUS IS DESIRED,
310& ASSIGN A VALUE",/,"OF 0.0 TO THIS VARIABLE.")
320 READ,RC
330 IF(RC.EQ.0.0) GO TO 20
340 RMIN=RC
350 Z=((VEL**2)*B)/(G*RMIN)
360 ARCL=15.*Z
365 20 ARCL1 = ARCL
370 PRINT 3,ARCL
380 3 FORMAT(/,"READ DESIGN ENTERING SPIRAL LENGTH. THE MINIMUM RECOM
390&MENDED",/,"SPIRAL LENGTH =",F10.2,"FT. IF MINIMUM LENGTH DESIRED,
400& ASSIGN A VALUE",/,"OF 0.0 TO THIS VARIABLE.")
410 READ,LS1
420 PRINT 4,ARCL
430 4 FORMAT(/,"READ DESIGN EXIT SPIRAL LENGTH. THE MINIMUM RECOM
440&MENDED",/,"SPIRAL LENGTH =",F10.2,"FT. IF MINIMUM LENGTH DESIRED,
450& ASSIGN A VALUE",/,"OF 0.0 TO THIS VARIABLE.")
460 READ,LS
470 IF(LS1.NE.0.0) ARCL=LS1
480 IF(LS.NE.0.0) ARCL1=LS
490 THETA=ATAN(Z/B)
500 IF(THETA.GT.10.) PRINT 25,THETA
510 25 FORMAT(/,"THE CROSS SLOPE OF THE WATER SURFACE = ",F5.2,"DEG.
520& THIS EXCEEDS THE",/,"RECOMMENDED MAXIMUM VALUE OF 10 DEGREES.")
530 PRINT 5
540 5 FORMAT(/,"IS FREE DRAINAGE DESIRED? TYPE YES OR NO ")
550 READ 11,JJ
560 11 FORMAT(A3)
570 IF(JJ.EQ.2HNO) GO TO 40
580 ARCL2 = (1.01*Z)/(2.*S)
590 IF(ARCL1.GT.ARCL2) GO TO 35
600 PRINT 10,ARCL2
610 10 FORMAT(/,"MUST INCREASE EXIT SPIRAL LENGTH TO",F8.2,"FT. TO
620& PROVIDE FREE DRAINAGE.")
630 ARCL1=ARCL2
640 35 PRINT 6
650 6 FORMAT(/,"ARE EQUAL SPIRAL LENGTHS DESIRED? TYPE YES OR NO!")
660 READ 11,JK
670 IF(JK.EQ.3HYES) ARCL=ARCL1
680 40 CALL PRT1(Q,S,B,KS,TEMP,ANG,YC,HYRAD,CHZC1,VEL,FRUDN,REYN,
681&RELRO,Z)
690 IF(Z.LT.0.5) GO TO 100
700 IF(FRUDN.LT.1.0) GO TO 100
710 CALL SPIRAL(ARCL,ARCL1,RMIN,ANG,Z)
720 100 STOP
730 END
1000 SUBROUTINE HYD(Q,S,B,KS,KV,ANG,YC,HYRAD,CHZC1,VEL,FRUDN,REYN,
1001&RELRO,Z,RMIN,ARCL)
1005 REAL KS,KV
1010 G=32.2
1020 YC=((Q**2.)/(G*(B**2.)))**(1./3.)
1030 12 CHZC = (Q/(B*YC))*SQRT((B+(2.*YC))/(S*B*YC))
1040 A=(CHZC*KV*(B+(2.*YC)))/(20.8*Q)
1050 A1=(KS*(B+(2.*YC)))/(12.8*YC*B)
1060 C=A+A1
1070 CHZC1 = -32.6*ALOG10(C)
1080 D = CHZC1-CHZC
1090 E = (CHZC1+CHZC)/2.
2000 DIFF = D/E
2010 IF(DIFF) 8,21,11
2030 8 IF(DIFF.GT.-0.9999) GO TO 21
2040 YC = YC+0.01
2050 GO TO 12
2060 11 IF(DIFF.LT.0.0001) GO TO 21
2070 YC = YC-0.01
2080 GO TO 12
2090 21 HYRAD = (YC*B)/(B+2.*YC)
2100 AREA = YC*B
2110 VEL = Q/AREA
2120 VELH = (VEL**2.)/(2.*G)
2130 REYN = (4.*VEL*HYRAD)/KV
2140 FRUDN = VEL/SQRT(G*YC)
2150 RELRO = HYRAD/KS
2160 RMIN=4.*FRUDN**2*B
2170 Z = ((VEL**2.)*B)/(G*RMIN)
2180 ARCL = 15.*Z
2181 RETURN
2182 END
2185 SUBROUTINE PRT1(Q,S,B,KS,TEMP,ANG,YC,HYRAD,CHZC1,VEL,FRUDN,REYN,
2186&RELRO,Z)
2187 REAL KS,KV
2190 PRINT 17
2200 17 FORMAT(/,20X,"HYDRAULIC ELEMENTS OF DESIGN",//,25X,"GIVEN
2210& DESIGN DATA",//,3X,"Q,CFS",4X,"SLOPE",5X,"WIDTH",7X,
2220&"KS",4X,"WATER TEMP.",7X,"DEF.ANGLE")
2230 PRINT 23,Q,S,B,KS,TEMP,ANG
2240 23 FORMAT(/,1X,F7.0,3X,F5.4,5X,F6.2,4X,F5.4,6X,F5.1,13X,F6.3)
2250 PRINT 24
2260 24 FORMAT(/,20X,"COMPUTED HYDRAULIC PARAMETERS",//,2X,"DEPTH
2270&HYD RAD   CHEZY C      VEL     FR.#       REY.#       R/KS")
2280 PRINT 27,YC,HYRAD,CHZC1,VEL,FRUDN,REYN,RELRO
2290 27 FORMAT(/,2X,F5.2,5X,F5.2,4X,F6.2,4X,F7.2,3X,F4.2,1X,E10.4,
2300&1X,F8.2)
2310 IF(FRUDN.LE.1) PRINT 28
2320 IF(Z.LT.0.5) PRINT 29
2330 28 FORMAT(/,"SUBCRITICAL FLOW - BANKING NOT REQUIRED",/)
2340 29 FORMAT(/,"BANKING LESS THAN 0.5 FT. - BANKING AND SPIRALS NOT
2341& REQUIRED",/)
2350 RETURN
2360 END
3000 SUBROUTINE SPIRAL(ARCL,ARCL1,RMIN,ANG,Z)
3020 DIMENSION N(2,100),STAS(2,100),DELTA(2,100),X(2,100),Y(2,100),
3030&R(2,100),MDG(2,100),MIN(2,100),MSEC(2,100),NUM(2),ARC(2),NDELTA
3031&(2),DELTAN(2,100),C(2,100),DELTA1(2),H(2),O(2),TL(2),DEFC(50),
3032&STA(50),ISTA(50),STAF(50),MDGC(50),MINC(50),MSECC(50),ISTAS(2,50),
3033&STASF(2,50),CHORDC(50)
3034 DATA N(1,1),N(2,1),MDG(1,1),MDG(2,1),MIN(1,1),MIN(2,1),MSEC(1,1)
3035&,MSEC(2,1)/8*0/,DELTA(1,1),DELTA(2,1),X(1,1),X(2,1),Y(1,1),Y(2,1),
3036&C(1,1),C(2,1)/8*0.0/
3040 ARC(1)=ARCL
3050 ARC(2)=ARCL1
3070 NN=2
3100 20 DO 30 I=1,NN
3105 ANUM=ARC(I)/12.5
3110 NUM(I)=ARC(I)/12.5
3112 DIFNUM=ANUM-NUM(I)
3114 IF(DIFNUM.GT.0.) NUM(I)=NUM(I)+1
3120 NDELTA(I)=2578310.078/((2.*NUM(I)-1.)*RMIN)
3130 30 ARC(I)=NUM(I)*12.5
3240 DO 60 I=1,NN
3250 NUM(I)=NUM(I)+1
3257 LL=NUM(I)
3260 DO 50 J=2,LL
3270 N(I,J)=N(I,J-1)+1
3280 STAS(I,J)=12.5*N(I,J)
3282 ISTAS(I,J)=STAS(I,J)/100
3283 STASF(I,J)=STAS(I,J)-(ISTAS(I,J)*100)
3290 DELTA(I,J)=(2.*N(I,J)-1.)*NDELTA(I)
3295 DELTAN(I,J)=N(I,J)**2*NDELTA(I)-DELTA(I,J)/2.
3300 R(I,J)=2578310.078/((2.*N(I,J)-1.)*NDELTA(I))
3310 TH=DELTAN(I,J)/206264.8062
3320 C(I,J)=2.*R(I,J)*SIN(DELTA(I,J)/206264.8062/2.)
3330 X(I,J)=X(I,J-1)+C(I,J)*COS(TH)
3340 Y(I,J)=Y(I,J-1)+C(I,J)*SIN(TH)
3345 ARG1=Y(I,J)/X(I,J)
3350 A=ATAN(ARG1)*206264.8062
3360 MDG(I,J)=A/60/60
3370 MIN(I,J)=(A-MDG(I,J)*60*60)/60
3380 MSEC(I,J)=A-MDG(I,J)*60*60-MIN(I,J)*60
3400 50 CONTINUE
3402 DELTA1(I)=(N(I,LL)**2*NDELTA(I))/3600.
3404 H(I)=X(I,LL)-RMIN*SIN(DELTA1(I)/57.2958)
3406 O(I)=Y(I,LL)-RMIN*(1.-COS(DELTA1(I)/57.2958))
3408 60 CONTINUE
3415 TL(1)=((RMIN+O(2))-(RMIN+O(1))*COS(ANG/57.2958))/SIN(ANG/57.2958)
3416&+H(1)
3420 TL(2)=((RMIN+O(1))-(RMIN+O(2))*COS(ANG/57.2958))/SIN(ANG/57.2958)
3421&+H(2)
3460 75 DELTAC=ANG-(DELTA1(1)+DELTA1(2))
3470 CLGTH=(DELTAC*RMIN)/57.2958
3475 SCC=ARC(1)
3480 CSS=SCC+CLGTH
3490 ICSS=CSS/100
3500 CSSF=CSS-(ICSS*100)
3510 DEGC=5729.58/RMIN
3520 ISCC=SCC/100
3530 SCCF1=SCC-(ISCC*100)
3540 ICLGTH=CLGTH/100
3550 CLGTHE=CLGTH-(ICLGTH*100)
3560 CLGTHG=100.-CLGTHE
3570 IF(SCCF1.GE.CLGTHG) GO TO 80
3580 NN=CLGTH/100
3590 GO TO 85
3600 80 NN=CLGTH/100+1
3610 85 CHORDC(1)=100.-SCCF1
3620 DEFC(1)=0.3*CHORDC(1)*DEGC
3630 STA(1)=SCC+CHORDC(1)
3640 ISTA(1)=STA(1)/100
3650 STAF(1)=STA(1)-(ISTA(1)*100)
3660 MDGC(1)=DEFC(1)/60
3670 MINC(1)=DEFC(1)-MDGC(1)*60
3680 MSECC(1)=DEFC(1)*60-MDGC(1)*3600-MINC(1)*60
3690 DO 90 I=2,NN
3700 CHORDC(I)=100.
3710 DEFC(I)=0.3*CHORDC(I)*DEGC+DEFC(I-1)
3720 STA(I)=STA(I-1)+100.
3730 ISTA(I)=STA(I)/100
3740 STAF(I)=STA(I)-(ISTA(I)*100)
3750 MDGC(I)=DEFC(I)/60
3760 MINC(I)=DEFC(I)-MDGC(I)*60
3780 MSECC(I)=DEFC(I)*60-MDGC(I)*3600-MINC(I)*60
3790 90 CONTINUE
3800 K=NN+1
3810 CHORDC(K)=CSS-STA(NN)
3820 DEFC(K)=0.3*CHORDC(K)*DEGC+DEFC(K-1)
3830 MDGC(K)=DEFC(K)/60
3840 MINC(K)=DEFC(K)-MDGC(K)*60
3850 MSECC(K)=DEFC(K)*60-MDGC(K)*3600-MINC(K)*60
3860 STA(K)=CSS
3870 ISTA(K)=ICSS
3880 STAF(K)=CSSF
3890 STT=CSS+ARC(2)
3900 ISTT=STT/100
3910 STTF=STT-(ISTT*100)
3915 KK=NUM(2)
3916 MM=NUM(1)
3930 DO 95 J=1,KK
3940 STAS(2,J)=STT-(N(2,J)*12.5)
3950 ISTAS(2,J)=STAS(2,J)/100
4000 95 STASF(2,J)=STAS(2,J)-(ISTAS(2,J)*100)
4010 BDEF=DELTAN(1,MM)+DELTA(1,MM)/2.-ATAN(Y(1,MM)/X(1,MM))*206264.8062
4020 MDGB=BDEF/3600
4030 MINB=(BDEF-MDGB*3600)/60
4040 MSECB=BDEF-MDGB*3600-MINB*60
4050 IARC=ARC(1)/100
4060 ARCF=ARC(1)-(IARC*100)
4070 IARC1=ARC(2)/100
4080 ARC1F=ARC(2)-(IARC1*100)
4090 PRINT 100,RMIN,Z
4100 100 FORMAT(//,25X,"CHANNEL CURVE ELEMENTS",//,"RADIUS OF CURVATURE
4110&(FT) = ",F10.3,/,"BANKING (FT) = ",F10.3)
4120 PRINT 101,IARC,ARCF,NDELTA(1),ISCC,SCCF1,ICLGTH,CLGTHE,ICSS,CSSF,
4121&IARC1,ARC1F,NDELTA(2),ISTT,STTF
4130 101 FORMAT(//,10X,"T.S. = STA 00 + 00.00",/,21X,I2," + ",F5.2,
4131&" = LENGTH OF ENTERING SPIRAL #",I4,/,21X,"-----------",/,10X,
4132&"S.C. = STA ",I2," + ",F5.2,/,21X,I2," + ",F5.2," = LENGTH OF
4133& CIRCULAR CURVE",/,21X,"-----------",/,10X,"C.S. = STA ",I2," + "
4134&,F5.2,/,21X,I2," + ",F5.2," = LENGTH OF EXIT SPIRAL #",I4,/,21X,
4135&"-----------",/,10X,"S.T. = STA ",I2," + ",F5.2," = TOTAL LENGTH
4136& OF CURVE")
4140 PRINT 102,DELTA1(1),DELTA1(2),TL(1),TL(2)
4150 102 FORMAT(//,"CENTRAL ANGLE OF ENTERING SPIRAL(DEG) = ",F10.4,/,
4151&"CENTRAL ANGLE OF EXIT SPIRAL(DEG) = "F10.4,/,"TOTAL ENTERING
4152& TANGENT LENGTH(FT) = ",F10.2,/,"TOTAL EXIT TANGENT LENGTH = ",
4153&F10.2)
4160 IF(ARC(1).NE.ARC(2)) PRINT 104,X(1,MM),Y(1,MM),X(2,KK),Y(2,KK)
4165 IF(ARC(1).EQ.ARC(2)) PRINT 103,X(1,MM),Y(1,MM)
4180 103 FORMAT(/,10X,"TANGENT OFFSET COORDINATES OF S.C. AND C.S.
4181& FROM",/,10X,"T.S. AND S.T. RESPECTIVELY",/,24X,"X(FT)",12X,
4182&"Y(FT)",/,20X,F8.2,10X,F8.2)
4190 104 FORMAT(/,10X,"TANGENT OFFSET COORDINATES OF S.C. FROM T.S.",
4191&/,24X,"X(FT)",12X,"Y(FT)",/,20X,F8.2,10X,F8.2,//,10X,"TANGENT
4192& OFFSET COORDINATES OF C.S. FROM S.T.",/,24X,"X(FT)",12X,"Y(FT)",
4193&/,20X,F8.2,10X,F8.2)
4200 105 PRINT 106
4310 106 FORMAT(/,24X,"CENTERLINE CURVE LAYOUT",/,7X,"DEFLECTION ANGLES
4311& FOR ENTERING SPIRAL WITH TRANSIT AT T.S.",/,4X,"DEFLECTION ANGLE
4312&  STATION    CHORD LENGTH   COORDINATES",/,6X,"DEG MIN SEC",19X,
4313&"(FT)",8X,"X(FT)     Y(FT)",/,2X,"TRANSIT AT T.S. = 00 + 00.00",
4314&6X,"00.000      00.000   00.000")
4320 PRINT 108,(MDG(1,J),MIN(1,J),MSEC(1,J),ISTAS(1,J),STASF(1,J),
4321&C(1,J),X(1,J),Y(1,J),J=2,MM)
4330 108 FORMAT(7X,I2,2X,I2,2X,I2,3X,I2," + ",F5.2,4X,F8.3,3X,F8.3,
4331&1X,F8.3)
4340 PRINT 110,MDGB,MINB,MSECB,ISCC,SCCF1
4350 110 FORMAT(/,4X,"DEFLECTION ANGLES FOR CIRCULAR CURVE WITH TRANSIT
4351& AT THE S.C.",//,10X,"BACK DEFLECTION TO T.S.= ",I2,"(DEG) ",I2,
4352&"(MIN) ",I2,"(SEC)",//,18X,"DEFLECTION ANGLE",11X,"STATION",/,
4353&22X,"DEG MIN SEC",/,27X,"TRANSIT AT S.C. = ",I2," + ",F5.2)
4360 PRINT 112,(MDGC(I),MINC(I),MSECC(I),ISTA(I),STAF(I),I=1,K)
4370 112 FORMAT(23X,I2,2X,I2,2X,I2,12X,I2," + ",F5.2)
4380 PRINT 114,ISTT,STTF
4390 114 FORMAT(/,9X,"DEFLECTION ANGLES FOR EXIT SPIRAL WITH TRANSIT
4391& AT S.T.",/,9X,"(SPIRAL RUN BACKWARDS)",//,4X,"DEFLECTION ANGLE
4392&  STATION    CHORD LENGTH   COORDINATES",/,6X,"DEG MIN SEC",19X,
4393&"(FT)",8X,"X(FT)     Y(FT)",/,2X,"TRANSIT AT T.S. = ",I2," + ",
4394&F5.2,6X,"00.000      00.000   00.000")
4400 PRINT 108,(MDG(2,J),MIN(2,J),MSEC(2,J),ISTAS(2,J),STASF(2,J),
4401&C(2,J),X(2,J),Y(2,J),J=2,KK)
4405 RETURN
4410 END
```

PROGRAM LISTING

PLATE D-2



COMPUTER PROGRAM FLOW CHART

PLATE D-3



COMPUTER PROGRAM
HYDRAULIC SUBROUTINE

PLATE D-4



PLATE D-5

EXAMPLE PROBLEM:

GIVEN: Q = 15,000 CFS S = 0.01 FT/FT B = 50.0 FT
KS = 0.002 FT WATER TEMP = 60° F
ANG = 45°

FIND: SHORTEST CURVE FOR GIVEN CONDITIONS AND PROVIDE FREE DRAINAGE.

```
READ DESIGN DATA - Q,S,B,KS,TEMP,ANG
INPUT:00250
 ?15000.,.01,50.,.002,60.,45.

READ THE DESIGN RADIUS OF CURVATURE. THE MINIMUM RECOMMENDED
RADIUS =    1049.44FT. IF MINIMUM RADIUS IS DESIRED, ASSIGN A VALUE
OF 0.0 TO THIS VARIABLE.
INPUT:00320
 ?0.0

READ DESIGN ENTERING SPIRAL LENGTH. THE MINIMUM RECOMMENDED
SPIRAL LENGTH =      30.40FT. IF MINIMUM LENGTH DESIRED, ASSIGN A VALUE
OF 0.0 TO THIS VARIABLE.
INPUT:00410
 ?0.0

READ DESIGN EXIT SPIRAL LENGTH. THE MINIMUM RECOMMENDED
SPIRAL LENGTH =      30.40FT. IF MINIMUM LENGTH DESIRED, ASSIGN A VALUE
OF 0.0 TO THIS VARIABLE.
INPUT:00460
 ?0.0

IS FREE DRAINAGE DESIRED? TYPE YES OR NO
INPUT:00550
 ?YES

MUST INCREASE EXIT SPIRAL LENGTH TO  102.34FT. TO PROVIDE FREE DRAINAGE.

ARE EQUAL SPIRAL LENGTHS DESIRED? TYPE YES OR NO!
INPUT:00660
 ?NO
```

PROGRAM INPUT EXAMPLE

PLATE D-6

```
                   HYDRAULIC ELEMENTS OF DESIGN

                        GIVEN DESIGN DATA

  Q,CFS     SLOPE       WIDTH        KS     WATER TEMP.          DEF.ANGLE

 15000.     .0100       50.00     .0020        60.0                45.000

                   COMPUTED HYDRAULIC PARAMETERS

 DEPTH      HYD RAD    CHEZY C       VEL     FR.#       REY.#       R/KS

  8.11        6.12     149.52       37.01    2.29 0.7527E+08    3060.70


                        CHANNEL CURVE ELEMENTS

RADIUS OF CURVATURE(FT) =    1049.436
BANKING (FT) =        2.027


          T.S. = STA 00 + 00.00
                      0 + 37.50 = LENGTH OF ENTERING SPIRAL # .491
                     -----------
          S.C. = STA  0 + 37.50
                      7 + 42.40 = LENGTH OF CIRCULAR CURVE
                     -----------
          C.S. = STA  7 + 79.90
                      1 + 12.50 = LENGTH OF EXIT SPIRAL # 144
                     -----------
          S.T. = STA  8 + 92.40 = TOTAL LENGTH OF CURVE


CENTRAL ANGLE OF ENTERING SPIRAL(DEG) =        1.2275
CENTRAL ANGLE OF EXIT SPIRAL(DEG) =       3.2400
TOTAL ENTERING TANGENT LENGTH(FT) =       450.31
TOTAL EXIT TANGENT LENGTH =       487.45

          TANGENT OFFSET COORDINATES OF S.C. FROM T.S.
                        X(FT)              Y(FT)
                       37.50                0.28

          TANGENT OFFSET COORDINATES OF C.S. FROM S.T.
                        X(FT)              Y(FT)
                      112.46                2.13
```

```
                          CENTERLINE CURVE LAYOUT
       DEFLECTION ANGLES FOR ENTERING SPIRAL WITH TRANSIT AT T.S.
    DEFLECTION ANGLE  STATION     CHORD LENGTH    COORDINATES
      DEG MIN SEC                     (FT)          X(FT)      Y(FT)
  TRANSIT AT T.S. = 00 + 00.00       00.000        00.000     00.000
          0   4   5    0 + 12.50     12.500        12.500      0.015
          0  12  16    0 + 25.00     12.500        25.000      0.089
          0  25  54    0 + 37.50     12.500        37.498      0.283

    DEFLECTION ANGLES FOR CIRCULAR CURVE WITH TRANSIT AT THE S.C.

          BACK DEFLECTION TO T.S.=  0(DEG) 47(MIN) 44(SEC)

                  DEFLECTION ANGLE             STATION
                      DEG MIN SEC
                           TRANSIT AT S.C. =   0 + 37.50
                         1  42  22             1 +  0.00
                         4  26   9             2 +  0.00
                         7   9  56             3 +  0.00
                         9  53  44             4 +  0.00
                        12  37  31             5 +  0.00
                        15  21  19             6 +  0.00
                        18   5   6             7 +  0.00
                        20  15  58             7 + 79.90

         DEFLECTION ANGLES FOR EXIT SPIRAL WITH TRANSIT AT S.T.
         (SPIRAL RUN BACKWARDS)

    DEFLECTION ANGLE  STATION     CHORD LENGTH    COORDINATES
      DEG MIN SEC                     (FT)          X(FT)      Y(FT)
  TRANSIT AT T.S. =    8 + 92.40     00.000        00.000     00.000
          0   1  12    8 + 79.90     12.500        12.500      0.004
          0   3  35    8 + 67.40     12.500        25.000      0.026
          0   7  35    8 + 54.90     12.500        37.500      0.083
          0  13  11    8 + 42.40     12.500        49.999      0.192
          0  20  23    8 + 29.90     12.500        62.498      0.371
          0  29  11    8 + 17.40     12.500        74.995      0.637
          0  39  35    8 +  4.90     12.500        87.490      1.008
          0  51  35    7 + 92.40     12.500        99.980      1.501
          1   5  11    7 + 79.90     12.500       112.464      2.133
STOP

RUNNING TIME:   13.9 SECS   I/O TIME :   06.1 SECS
```

PROGRAM OUTPUT EXAMPLE

PLATE D-7



FREE DRAINAGE CRITERIA

PLATE D-8

ALTERNATIVE	RADIUS CURVE	CURVE BANKING	BOTH SPIRALS	SPIRAL LENGTH	FREE DRAINAGE	REMARKS
(1)	MIN EQ D-2	MAX EQ D-4	EQUAL	MIN EQ D-5	NO	SHORTEST TOTAL LENGTH
(2)	MIN EQ D-2	MAX EQ D-4	EQUAL	ARBITRARY > EQ D-5	NO	
(3)	ARBITRARY > EQ D-2	CORRESPONDING EQ D-3	EQUAL	CORRESPONDING EQ D-5	NO	
(4)	ARBITRARY > EQ D-2	CORRESPONDING EQ D-3	EQUAL	ARBITRARY > EQ D-5	NO	
(5)	ARBITRARY > EQ D-2	CORRESPONDING EQ D-3	UNEQUAL	ARBITRARY > EQ D-5	NO	
(6)	MIN EQ D-2	MAX EQ D-4	UNEQUAL	ENTERING = MIN EQ D-5 EXIT TO DRAIN > EQ D-5	YES	SIMILAR TO (1)
(7)	MIN EQ D-2	MAX EQ D-4	EQUAL	ENTER = TO EXIT EXIT TO DRAIN > EQ D-5	YES	SIMILAR TO (1)
(8)	ARBITRARY > EQ D-2	CORRESPONDING EQ D-3	UNEQUAL	ENTER = CORRESPONDING EQ D-5 EXIT TO DRAIN > EQ D-5	YES	SIMILAR TO (3)
(9)	ARBITRARY > EQ D-2	CORRESPONDING EQ D-3	EQUAL	ENTER = EXIT EXIT TO DRAIN > EQ D-5	YES	SIMILAR TO (3)
(10)	ARBITRARY > EQ D-2	CORRESPONDING EQ D-3	UNEQUAL	ENTER = ARBITRARY > EQ D-5 EXIT TO DRAIN > EQ D-5	YES	SIMILAR TO (4)
(11)	ARBITRARY > EQ D-2	CORRESPONDING EQ D-3	EQUAL	ENTER= EXIT EXIT TO DRAIN > EQ D-5	YES	SIMILAR TO (4)
(12)	ARBITRARY > EQ D-2	CORRESPONDING EQ D-3	UNEQUAL	ENTER = ARBITRARY > EQ D-5 EXIT TO DRAIN > EQ D-5	YES	SIMILAR TO (5)

SPIRAL ALTERNATIVES COMPARISONS

PLATE D-9